2018

Semester 2

Kemito Pipfruit: Optimisation

Connor McDowall
Josh Beckett
Alexander Zhao

September 10, 2018

Listings

Contents

List of Figures

List of Tables

1 Problem Description/Background

Kemito Pipfruit pack and distribute apples and avocados. They have a number of suppliers who provide produce, pack produce at Kemitos packhouses and ship the produce to various markets. Kemito wish to invest in new automated packing machines in their four packhouse locations. They wish to facilitate the transhipment of produce and meet the markets demand. Our objective is to decide the produce setting, size and number of packing machines to build at each packhouse. Our ancillary objective is to use optimisation to minimise the cost of produce transhipment from; supplier to packhouse and packhouse to market. The transhipment of produce and investment in machines for apples and avocados are mutually exclusive, therefore, can be treated as separate problems. In addition, demand varies at each market per period and is not known beforehand.

2 Data

The data given included: the fixed supply (units/period) for four avocado and ten apple producers, fixed per period. The historical, variable demand (units/period) for five avocado markets and fifteen apple markets for ten periods. The transportation costs per unit for apples and avocados from supplier to packhouse and packhouse to market. To conclude, the average packing rate (units/period) and cost (000/machine) of packing machine size (small, medium and large) completes the set of data.

The variable, historical demand for the twenty markets over the ten periods created uncertainty. The periods beginning, duration and correlation with other periods was unknown. These uncertainties created difficulties in formulating the model as there appeared to be no pattern per period or any indication of the likely cause.

We considered taking the peak value of each market demand across all periods but lead to a mass shortage of produce, unable to satisfy the demand of each market. Also, the cost of this solution would be exorbitant. Averaging the data across the periods was also considered. This resulted in the demand not being meet for several time periods while not considering fluctuating demand. We considered using a weighting system to penalise or omit unlikely periods, however, we did not have the industry expertise to deem what was an unlikely scenario. We agreed to use the data to build a robust solution by considering all periods.

3 Assumptions

We made the following assumptions to simplify our model formulation:

- Meeting market demand is a priority. This meant we solved our model to ensure that all the different market demands' for each period were met.
- Suppliers contracts must be honoured meaning we will not take more than what the producers can provide and we will not seek out contracts with others. The supply from each supplier is fixed for any period.
- No wastage at packhouses meaning produce flow is conserved. This may be unrealistic as human error, mechanical failure or transporation may create wastage.
- Minimising the cost of operation is our main driver. We are not concerned with the profitability of produce. We focus on the optimal locations for packing machines and the transportation of fruit between suppliers to packhouses and packhouses to markets.
- The location of packing machines is permanent. Machines cannot be decommissioned or transported to new locations. This ensures that our solution is very robust and can handle different levels of demand.

4 Model Formulation

Our model was formulated as a naturally integer linear programme, written in AMPL and solved using Gurobi. (Note: AMPL uses names for index notation rather than numbers).

4.1 Data

4.1.1 Sets and Parameters

Due the mutually exclusive nature of produce transhipment, two data files were defined from the data. A file for each fruit. Multiple sets were set in both files. These sets are the suppliers, periods, markets, pack machine sizes and packhouse locations. These sets function as objects to assign parameters to individual sets and/or a combination of sets. Arcs were created between suppliers to packhouses and packhouses to markets as an additional set. Each set was assigned relevant parameters. These parameters are the number of periods, the supply of each supplier. the demand of each market for each period, the pack rate for each pack machine size, the cost for each packing machine size, and the transportation costs between every supplier to every packhouse and every packhouse to every market. Arcs were also assigned lower and upper limits. These sets and parameters defined for the model can be found in the appendix (7.1).

4.2 Model

4.2.1 Variables

Flow and Built are the two decision variables. Flow is the number of units of produce shipped in the arc for a period. Built is the number of machines of each size built at the packhouse location. See the variables below (4.2.1).

- var $Flow_{ijp} \geq 0$, integer where i = origin in arc, j = destination in arc, p = period.
- var $Built_{mh} \geq 0$ where m = packmachine and h = packhouse.

4.2.2 Objective Function

Our objective function is to minimise the combined cost of installing the required number and size of packmachines at each packhouses, with transporting produce flow between arcs across all periods. See the function below (4.2.2).

$$\text{Min} \sum_i \sum_j \sum_p Cost_{ij} \times Flow_{ijp} + \sum_m \sum_h numPeriods \times packcost_m \times Built_{mh}$$

where i = origin, j = destination, p = period, m = packmachine, h = packhouse.

4.2.3 Constraints

Four constraints bind the model; Demand for all produce must be met at all markets. The total produce transported to packhouses must be less than or equal to supply. Aggregate flow into each packhouse must equal aggregate flow out of that packhouse, conserving the flows. Finally, the capacity of each packhouse's combined number of machines may not be exceeded by the flows in. The constraints are expressed mathematically below (4.2.3).

- Demand: $\sum_j Flow_{hjp} \geq demand_{jp}$
- Supply: $\sum_i Flow_{ihp} \leq supply_{ip}$
- Conserve: $\sum_i Flow_{ihp} = \sum_j Flow_{hjp}$
- Capacity: $\sum_m Built_{mh} \times rate_m \geq \sum_i Flow_{ihp}$

where i = supplier, j = market, h = packhouse, p = period and m = machine. See the whole AMPL Implementation of the model in 7.3 of the appendix.

5 Results

The machine investment plan explains the number and size of machines to install in each of your four packhouses. Table 1 outlines the proposed investment plan for both apples and avocados. The transhippment flows of apples and avocados varied due to the fluxuating demand at the markets across periods. The flows in each period were important to consider for machine installation but not the reported cost in the conclusions and recommendations 6. Future demand will likely be different. The transhippment flows for both produce in period one are displayed in the appendix 7.2.

Packhouse	Apple: Large	Apple: Medium	Avocado: Large	Avocado: Medium
One	-	1	-	-
Two	-	2	2	-
Three	2	-	-	3
Four	-	6	-	-

Table 1: Apple Machine Investment Plan

6 Conclusions

6.1 Recommendations

Based on the aforementioned results, Kemito Pipfruit should:

- Install one medium machine set to pack apples at Packhouse One.
- Install two medium and two large machines set to pack apples and avocados respectively at Packhouse Two.
- Install three medium and two large machines set to pack avocados and apples respectively at Packhouse Three.
- Install six medium machines set to pack apples at Packhouse Four.

The investment plan will cost **$440,000**. The model delivers a robust solution. Market demand is met in each period while minimising machine acquisition, installation and produce transhippment. See table 2 for the cost per unit for each machine size.

	Small	Medium	Large
Cost($)	10000	25000	35000

Table 2: Machine Size Cost

6.2 Improvements

We have delivered the best model based on the data you provided. With more data, we could formulate a model to provide a more robust solution. In particular:

- Using produce pricing to maximise the profit of your transhipment operations.
- Factoring in different product segments within apples and avocados.
- Factoring in produce wastage and conversion rates in transportation and packing.
- Use data to forecast period demand combined with potentially using futures contacts.
- Factoring in decommissioning and reinstalling packing machines in different packhouses.
- Using penalty costs for not meeting supply or demand, based on your existing contracts.

7 Appendix

7.1 Sets and Parameters

Sets which are assigned in the data file are defined in the model by:

- set SUPPLIERS;
- set MARKETS;
- set PERIODS;
- set PACKMACHINE;
- set ARCS := (SUPPLIERS cross PACKHOUSE) union (PACKHOUSE cross MARKETS);

7.1.1 Parameters

- param supply{SUPPLIERS};
- param demand{MARKETS};
- param rate{MARKEST,PERIODS};
- param packcost{PACKMACHINE};
- param supplycost{SUPPLIERS,PACKHOUSE};
- param marketcost{PACKHOUSE,MARKETS};
- param Cost{ARCS} default 0;
- param Lower{ARCS} $\geq$ 0;
- param $Upper_{ij} \geq Lower_{ij} \ \forall \ ARCS_{ij}$;
- param numPeriods;

7.2 Period One Transhippment Flows

Note: Both the avocado and apple transhippment flows vary per period.



Figure 1: Avocado Transhippment Flows: Period One



Figure 2: Apple Transhippment Flows: Period One

7.3 Model

```
# Optimites: OneFruityBoi
# Connor McDowall 530913386 cmcd398
# Josh  Beckett 528396260 jbec200
# Alexander Zhao 619051233 azha755
# Optimisation Model File

# Set all the parameters

set SUPPLIERS;
set MARKETS;
set PERIODS;
set PACKMACHINE;
set PACKHOUSE;

# Create a large set of ARCS
set ARCS := (SUPPLIERS cross PACKHOUSE) union (PACKHOUSE cross MARKETS);

# Set parameters

# Set the lower and upper bounds for all arcs
param Lower{ARCS} >=0, default 0;
param Upper{(i,j) in ARCS} >= Lower[i,j], default Infinity;

# Set all the parameters for Supply and Demand
param supply{SUPPLIERS};
param demand{MARKETS,PERIODS};
param rate{PACKMACHINE};
param packcost{PACKMACHINE};

# Do the cost tables and costs flows
```

```
param supplycost{SUPPLIERS,PACKHOUSE};
param marketcost{PACKHOUSE, MARKETS};
param Cost{ARCS} default 0;

# Set up the Number of Periods
param numPeriods;


# Set variables

# Create variables
# Three Dimensional System
var Flow {(i,j) in ARCS, p in PERIODS} >= 0, integer;

# Variable to control the number of machines to build.
var Built {PACKMACHINE, PACKHOUSE} >=0, integer;


# Objective Function

minimize TotalCost: sum{(i,j) in ARCS, p in PERIODS} Cost[i,j]*Flow[i,j,p]
+ sum{m in PACKMACHINE, h in PACKHOUSE} numPeriods*packcost[m]*Built[m,h];


# Constraints

# Ensure the Demand is met, meeting demand exactly
subject to MeetDemand {j in MARKETS, p in PERIODS}:
  sum {i in PACKHOUSE} Flow[i, j, p] >= demand[j,p];

# Ensure that supply is not breached
subject to UseSupply {i in SUPPLIERS, p in PERIODS}:
  sum {j in PACKHOUSE} Flow[i ,j ,p] <= supply[i];

# Equal flow constraint
subject to ConserveFlow {j in PACKHOUSE, p in PERIODS}:
   sum {i in SUPPLIERS} Flow[i, j, p] = sum{i in MARKETS} Flow[j, i, p];

# Not exceed capacity at packhouse for each period
subject to CapacityOut {h in PACKHOUSE, p in PERIODS}:
sum {m in PACKMACHINE} Built[m, h]*rate[m] >=sum {j in SUPPLIERS} Flow[j, h, p];


# Model summary notes.

# The model works for both Apples and Avocados
# You can treat avocados and applesas two seperate problems.
# Use the relevant data file for the problem.
# Avocado and Apple packing machines are mutually exclusive.
# Don't need to take all the supply, we buy from the suppliers
# and incur transporation costs. We want to minimise our cost and wastage.
# We have have contracts to buy from other suppliers.
# We assume the supply will not exceed the demand based on the data you
# have given us.
# We have deemed it not necessary to have a dummy demand.
# We have a contract rate with the suppliers. We are not obligued to
# take all of the supply.
```